

# Phrase Structure

---

<Program> ::= **prog IDENTIFIER** <Block> **gorp EOF**

<Block> ::= (<Declaration>; | <Command>;)\*

<Declaration> ::= <Type> **IDENTIFIER**

<Type> ::= <SimpleType> | <CompoundType>

<SimpleType> ::= **int** | **boolean** | **string**

<CompoundType> ::= **map** [ <SimpleType>, <Type> ]

<Command> ::= <LValue> = <Expression>  
| <LValue> = <PairList>  
| **print** <Expression>  
| **println** <Expression>  
| **do** (<Expression>) <Block> **od**  
| **do** (<LValue> : [ **IDENTIFIER** , **IDENTIFIER** ] ) <Block> **od**  
| **if** (<Expression> ) <Block> **fi**  
| **if** (<Expression> ) <Block> **else** <Block> **fi**  
|  $\epsilon$

<LValue> ::= **IDENTIFIER** | **IDENTIFIER** [ <Expression> ]

<Pair> ::= [ <Expression> , <Expression> ]

<PairList> ::= { <Pair> ( , <Pair> )\* } | { }

<Expression> ::= <Term> (<RelOp> <Term>)\*

<Term> ::= <Elem> (<WeakOp> <Elem>)\*

<Elem> ::= <Factor> (<StrongOp> <Factor>)\*

<Factor> ::= <LValue> | **INTEGER\_LITERAL** | **BOOLEAN\_LITERAL** | **STRING\_LITERAL**  
| ( <Expression> ) | ! <Factor> | -<Factor> | <PairList>

<RelOp> ::= **OR** | **AND** | **EQUALS** | **NOT\_EQUALS** | **LESS\_THAN** |  
**GREATER\_THAN** | **AT\_MOST** | **AT\_LEAST**

<WeakOp> ::= **PLUS** | **MINUS**

<StrongOp> ::= **TIMES** | **DIVIDE**